
3.1 Das D‑Flipflop

Flipflops sind Schaltungen mit zwei stabilen Zuständen, die jeweils durch äußere Anregung aktiviert werden. Wir betrachten im folgenden das „D‑Flipflop“, welches die Besonderheit besitzt, dass es auf Änderungen des Eingangssignals nur solange reagiert, wie der Takt auf „high“ ist.

Weiterhin versuchen wir, durch die Einführung der „Setup-Zeit“ tsu, der „Hold-Zeit“ th und den „Verzögerungszeiten“ tpLH bzw. tpHL das etwas träge Verhalten einer reellen Schaltung nachzuempfinden.

Bei dem D‑Flipflop handelt es sich um ein getaktetes Flipflop. Die halbe Länge einer Taktdauer (also genau ein „Berg“ oder ein „Tal“) nennen wir Thalbe.

Die Ports unseres D‑Flipflops bezeichnen wir mit „d“ (Eingang), „c“ (Takt), „g“ (Ausgang) und „gn“ (invertierter Ausgang).

3.1.1 Verhaltensbeschreibung

Unsere erste Aufgabe besteht darin, das Verhalten eines D‑Flipflop in einer Entity zu definieren:

library IEEE;

use IEEE.std_logic_1164.all;

entity D_FlipFlop is

  generic (tsu,th,tphl,tplh,thalbe : time);

  port (d,c : IN STD_LOGIC;

        q,qn: OUT STD_LOGIC);

end D_FlipFlop;

architecture Behaviour of D_FlipFlop is

  signal Flag : STD_LOGIC;

begin

  Ausgang: process -- keine sensitivity list, da wait/until-konstrukt

  begin

       wait until (c='1') or (d'event); -- Auf Takt- oder D-Wechsel warten

       -- Änderung erfolgt nur, wenn d gewechselt, Takt auf 1 u. Flag auf 0

       if (d/=d'last_value) and (c='1') and (Flag='0') then 

          if (d='1') then -- Q: 0 -> 1

             wait for tplh; 

          else -- d='0'

             wait for tphl; -- Q: 1 -> 0

          end if;

          q <= d;

          qn <= not d;

       end if;

  end process;  

  Validation: process

  begin

       wait until (c'event); -- clock muß wechseln -> Flanke

       if (c='0') then

          Flag <= '1'; -- q (Ausgang) verriegeln

          wait for th;

          Flag <= '0';

       else -- c='1'

          Flag <= '0'; -- q (Ausgang) freigeben

          wait for (thalbe-tsu);

          Flag <= '1';       

       end if;  

  end process;  

end Behaviour;

3.1.2 Stimuli-Datei

Um das definierte Verhalten des D‑Flipflop auf Korrektheit zu testen, erzeugen wir eine Simulationsdatei, die vor allem die kritischen Situationen simuliert, d.h. Signalwechsel während der „trägen“ Zeiten des Flipflop. Dazu zählt vor allem ein Wechsel des Eingangssignals unmittelbar vor und nach der fallenden Taktflanke (das Flipflop darf auf diesen Wechsel aufgrund der Setup-Zeit und der Hold-Zeit nicht reagieren).

Die konkreten Werte für das Zeitverhalten des Flipflop sind jetzt bekannt und können dem Quelltext entnommen werden.

Wir haben die Simulationsdatei wie folgt kodiert:

library IEEE;

use IEEE.std_logic_1164.all;

entity d_flipflop_stim is

end d_flipflop_stim;

architecture Test of d_flipflop_stim is

  component d_flipflop

    generic (tsu,th,tphl,tplh,thalbe : time);

    port (d,c : IN STD_LOGIC;

          q,qn: OUT STD_LOGIC);
  

  end component;

  component clock

    port (c      : OUT STD_LOGIC);

  end component;

  signal  d_test, c_test, q_test, qn_test: STD_LOGIC;

begin

  d_flipflop_test: d_flipflop generic map(

  


         20 ns,





 5 ns,





 10 ns,





 15 ns,





 50 ns




      ) 




      port map (




         d_test,





 c_test,





 q_test,





 qn_test




      );

  clock_test: clock port map (c_test);

  d_test <= '0' after 0 ns,


    '1' after 90 ns,   -- 10 ns vor fallender Flanke


    '0' after 202 ns,  -- 2 ns nach der fallenden Flanke


    '1' after 275 ns,  -- 25 ns nach steigender Flanke


    '0' after 375 ns;  -- 25 ns nach steigender Flanke 

end Test;

3.1.3 Konfiguration

configuration d_flipflop_config of d_flipflop_stim is

  for Test

    for d_flipflop_test: d_flipflop

      use entity work.d_flipflop( Behaviour );

    end for;

    for clock_test: clock

      use entity work.clock ( Behaviour2 );

    end for;

  end for;

end d_flipflop_config;

3.1.4 Simulation

Die Simulation des D‑Flipflop mittels der obigen Stimuli-Datei ergibt die folgende Ausgabe:

Abbildung 1
Die umkreisten Stellen zeigen die oben angesprochenen, „kritischen“ Übergänge. man sieht hier sehr deutlich, dass die Setup-Zeit und die Hold-Zeit die Dauer der „aktiven Phase“ des Flipflops auf eine Zeit Taktiv ( Thalbe beschränkt. Oder anders formuliert: Das Ausgangssignal des D‑Flipflop kann sich zwar nicht immer ändern, wenn der Takt gleich 1 ist, aber es kann sich auf keinen Fall dann ändern, wenn der Takt gleich 0 ist.

3.2 Das JK‑Flipflop

Wir betrachten nun das JK‑Flipflop, welches einen gravierenden Nachteil des D‑Flipflops ausräumt. An dem Simulationsergebnis des D‑Flipflop sieht man sehr deutlich, dass (solange der Takt auf 1 ist) der Ausgang des Flipflop dem Eingang unmittelbar folgt (abgesehen von den „Verzögerungszeiten“ tpLH bzw. tpHL.

Diese Situation ist nicht hinnehmbar unter dem Aspekt, dass bei der Hintereinanderschaltung mehrerer solcher Flipflops mit großer Wahrscheinlichkeit am Ende der Kette keine brauchbaren Ergebnisse mehr ankommen werden. das JK‑Flipflop bietet hier eine elegante Lösung.

Ein JK‑Flipflop besteht streng genommen aus zwei einzelnen Flipflops (die als „Master“ und „Slave“ bezeichnet werden). Der Clou ist, dass die beiden Flipflops ein zueinander invertiertes Taktsignal enthalten, so dass sie abwechselnd (mal das eine, mal das andere, aber nie beide zugleich) auf ihre jeweiligen Eingangssignale reagieren.

In dem vorangegangenen Experiment haben wir festgestellt, dass sich das Ausgangssignal des D‑Flipflop auf keinen Fall ändern kann, wenn der Takt gleich 0 ist. Daraus folgt, dass wir bei der Hintereinanderschaltung zweier D‑Flipflops die absolute Sicherheit haben, dass sich das Ausgangssignal des „ersten“ Flipflop nicht ändert, solange das „zweite“ Flipflop auf Änderungen des Eingangssignals sensibel reagiert.

Das JK‑Flipflop verfügt zusätzlich über eine andere Terminologie bezüglich der Eingänge, man spricht hier von „j“ und „k“. Es hat zwei Eingänge, weil es neben den „Befehlen“ 1 und 0 (kodiert durch j und k) auch die Befehle „halten“ und „invertieren“ kennt (sowohl j als auch k auf 0 bzw. 1).

3.2.1 Verhaltensbeschreibung

Im Gegensatz zu dem D‑Flipflop müssen wir hier keine besonderen Zeiten beachten, weil wir dem JK‑Flipflop ein „perfektes Verhalten“ unterstellen. Wir erzeugen das JK‑Flipflop aber keineswegs aus zwei autarken D‑Flipflops, sondern ahmen das Master-Slave-Verhalten mit Hilfe einer internen Variablen „g_intern“ nach.

Wir erzeugen nun (auf der geistigen Grundlage des Verhaltens des D‑Flipflop) das Verhalten eines JK‑Flipflop:

library IEEE;

use IEEE.std_logic_1164.all;

entity JK_FlipFlop is

  port (j,k,c : IN STD_LOGIC;

       q,qn: OUT STD_LOGIC);

end JK_FlipFlop;

architecture Behaviour of JK_FlipFlop is

signal q_intern : STD_LOGIC;

begin

  Ausgang: process -- keine sensitivity list, da wait/until-konstrukt

  begin

       wait until (c'event); -- Auf Taktflanke warten

       -- Fallunterscheidung, ob Takt steigt oder faellt

       if (c='1') then

           -- 1. Fall: loeschen

           if (j='0') and (k='1') then


      q_intern <= '0';


   end if;

           -- 2. Fall: setzen

           if (j='1') and (k='0') then


      q_intern <= '1';


   end if;

           -- 3. Fall: invertieren

           if (j='1') and (k='1') then


      q_intern <= not q_intern;


   end if;

       else

           q <= q_intern;


   qn <= not q_intern;

       end if;

  end process;  

end Behaviour;

3.2.2 Stimuli-Datei

Wir testen das JK‑Flipflop anhand willkürlich gewählter Zeitwerte:

library IEEE;

use IEEE.std_logic_1164.all;

entity jk_flipflop_stim is

end jk_flipflop_stim;

architecture Test of jk_flipflop_stim is

  component jk_flipflop

    port (j,k,c : IN STD_LOGIC;

          q,qn: OUT STD_LOGIC);
  

  end component;

  component clock

    port (c      : OUT STD_LOGIC);

  end component;

  signal  j_test, k_test, c_test, q_test, qn_test: STD_LOGIC;

begin

  jk_flipflop_test: jk_flipflop port map (




         j_test,





 k_test,





 c_test,





 q_test,





 qn_test




      );

  clock_test: clock port map (c_test);

  j_test <= '0' after 0 ns,


    '1' after 120 ns,   


    '0' after 220 ns,  


    '1' after 320 ns,  


    '0' after 420 ns,  


    '1' after 520 ns,  


    '0' after 620 ns,  


    '1' after 720 ns,  


    '0' after 820 ns,  


    '1' after 920 ns,  


    '0' after 1020 ns;  

  k_test <= '1' after 0 ns,


    '1' after 120 ns,   


    '0' after 220 ns,  


    '0' after 320 ns,  


    '1' after 420 ns,  


    '1' after 520 ns,  


    '0' after 620 ns,  


    '0' after 720 ns,  


    '0' after 820 ns,  


    '0' after 920 ns,  


    '0' after 1020 ns;  

end Test;

3.2.3 Konfiguration

configuration jk_flipflop_config of jk_flipflop_stim is

  for Test

    for jk_flipflop_test: jk_flipflop

      use entity work.jk_flipflop( Behaviour );

    end for;

    for clock_test: clock

      use entity work.clock ( Behaviour2 );

    end for;

  end for;

end jk_flipflop_config;

3.2.4 Simulation

Die Simulation des JK‑Flipflop mittels der obigen Stimuli-Datei ergibt die folgende Ausgabe:

Abbildung 2
3.3 Synchron sequentielle Schaltungen

3.3.1 Zustandstabelle

Da insgesamt sechs verschiedene Zustände dargestellt werden sollen, muß das Ausgaberegister log26 = 2,585... Bits haben. Da keine „krummen“ Bits existieren, wird die nächst größere ganze Zahl gewählt, in diesem Fall die 3. Zur Darstellung der 6 Zustände werden also 3 Bits im Ausgaberegister benötigt, die durch 3 JK-Flipflops repräsentiert werden.

Für den Wechsel von einem Zustand in einen anderen, müssen die Flipflops richtig beschaltet werden. Diese Beschaltung ist leicht zu ermitteln, da die Zustände für die Ausgänge durch die sechs Zustände binär codiert bereits vorgegeben sind. Für den Wechsel eines Zustandes in den in der Reihenfolge nachfolgenden, müssen dann die Ausgänge der Flipflops verändert werden, was durch Set/Reset/Hold und Toggle eines Flipflops erfolgt. Dadurch ergibt sich für die einzelnen Zustände der sequentiellen Schaltung folgende Zustandstabelle:


z
q2
q1
q0
q2‘
q1‘
q0‘
J2
K2
J1
K1
J0
K0

3
1
0
1
1
1
0
0
0
x
x
1
x
1

0
1
0
0
0
1
1
1
1
x
0
x
0
x

4
1
1
0
0
0
1
1
x
1
1
x
0
x

2
1
0
1
0
1
0
1
1
x
x
0
1
x

7
1
1
1
1
0
0
0
x
0
x
1
x
0

5
1
1
0
1
0
1
0
x
1
1
x
x
0

3
0
0
1
1
1
0
0
0
x
0
x
0
x

0
0
0
0
0
1
1
1
0
x
0
x
0
x

4
0
1
0
0
0
1
1
0
x
0
x
0
x

2
0
0
1
0
1
0
1
0
x
0
x
0
x

7
0
1
1
1
0
0
0
0
x
0
x
0
x

5
0
1
0
1
0
1
0
0
x
0
x
0
x

x bedeutet dabei don’t care, die Ansteuerung des entsprechenden Flipflop-Eingangs ist demnach egal, während bei 1 gesetzt und bei 0 die Beschaltung vorgegeben ist. Auffällig ist noch, dass im unteren Bereich der Tabelle für den Zustand z = 0 (also Anhalten des Zählvorgangs) die Flipflops ebenfalls den aktuellen Wert halten sollen, was der Ansteuerung J = 0 und K = x (don’t care) entspricht.

3.3.2 Ansteuertabelle

Um die konkrete „Verdrahtung“ der drei benötigten JK-Flipflops zu ermitteln, überträgt man die Information aus der oben angegebenen Zustandstabelle in 2 * 3 KV-Diagramme (je zwei Leitungen pro Flipflop). Damit werden dann die minimalen DNFs für die Ansteuerung der Bauelemente bestimmt.

Da sich die einzelnen Flipflops zunächst in einem undefinierten Zustand befinden, müssen sie mit Hilfe eines Flags r = 1 in den Startzustand (hier: 3 = 6. Zustand) gesetzt werden. Dazu verknüpft man die Flipflopsteuerleitungen noch mit dem r-Flag und erhält folgende Ansteuertabelle:

J2 = (z ( (q0) ( (r

K2 = (q2 ( (q1) ( r

J1 = (z ( q2) ( r

K1 = (z ( q0) ( (r

J0 = (z ( q1) ( r

K0 = (z ( (q2) ( (r

3.3.3 Verhaltensbeschreibung und Simulation

Mit diesen Informationen läßt sich jetzt der Zähler in VHDL modellieren. Dazu haben wir eine neue Entity counter und die entsprechenden Stimuli- und Configdateien geschrieben. An Komponenten haben wir den im vorherigen Abschnitt entworfenen JK-Flipflop und die gestellte Clock verwendet. Hier zunächst die Entity counter mit zugehöriger Architecture:

library IEEE;

use IEEE.std_logic_1164.all;

entity Counter is

  port (r,z,c       : IN STD_LOGIC;

        q0,q1,q2  : OUT STD_LOGIC);

end Counter;

architecture Behaviour of Counter is

  component jk_flipflop

    port (j,k,c : IN STD_LOGIC;

          q,qn: OUT STD_LOGIC);
  

  end component;

   signal q00,q11,q22,qn00,qn11,qn22,j0,j1,j2,k0,k1,k2 : STD_LOGIC;

begin

  -- Zaehler benoetigt drei Flip-Flops

  jk_0 : jk_flipflop port map( j0,k0,c,q00,qn00 );

  jk_1 : jk_flipflop port map( j1,k1,c,q11,qn11 );

  jk_2 : jk_flipflop port map( j2,k2,c,q22,qn22 );

  -- erstes Flip Flop definieren

  j0 <= (q11 and z) or r;

  k0 <= (qn22 and z) and (not r);

  -- zweites Flip Flop definieren

  j1 <= (q22 and z) or r;

  k1 <= (q00 and z) and (not r);

  -- drittes Flip Flop definieren

  j2 <= ((not q00) and z) and (not r);

  k2 <= (q22 and (not q11)) or r;

  -- Jetzt die Ausgabewerte der Flip Flops zurueckgeben

  q0 <= q00;

  q1 <= q11;

  q2 <= q22;

end Behaviour;

Die zugehörige Stimulidatei, in der die einzelnen Modi des Zählers getestet werden (Startzustand setzen r = 1, Zählen z = 1, Nicht Zählen z = 0):

library IEEE;

use IEEE.std_logic_1164.all;

entity counter_stim is

end counter_stim;

architecture Test of counter_stim is

  component clock

    port (c      : OUT STD_LOGIC);

  end component;

  component counter

    port (r,z,c       : IN STD_LOGIC;

          q0,q1,q2  : OUT STD_LOGIC);

  end component;

  signal  c_test, r_test, z_test, q0_test, q1_test, q2_test: STD_LOGIC;

begin

  -- wir benoetigen einen Counter ...

  counter_test: counter port map (




         r_test,





 z_test,





 c_test,





 q0_test,





 q1_test,





 q2_test




      );

  -- ... und einen Takt dafuer


    

  clock_test: clock port map (c_test);

  -- 'z' ist der Eingang der zum Anhalten fuehrt

  z_test <= '0' after 0 ns,


    '1' after 100 ns,    -- Zählen


    '0' after 1000 ns,   -- Nicht Zählen


    '1' after 1400 ns,   -- Wieder Zählen 


    '0' after 3000 ns;   -- Stop

  -- 'r' ist der Eingang der den Startzustand setzt
 

  r_test <= '1' after 0 ns,  -- Startzustand setzen (Zustand 6)


    '0' after 100 ns,    -- Schaltung nicht beeinflussen


    '1' after 2000 ns,   -- Startzustand erneut setzen...


    '0' after 2400 ns;   -- ...und erneut nicht beeinflussen

end Test;

Die zugehörige Configdatei enthält Bezüge zu den verwendeten Entities und Architectures:

configuration counter_config of counter_stim is

  for Test

    for counter_test: counter

      use entity work.counter( Behaviour );

         for Behaviour

            for all : jk_flipflop


       use entity work.jk_flipflop( Behaviour );


    end for;


 end for;

    end for;

    for clock_test: clock

      use entity work.clock ( Behaviour2 );

    end for;

  end for;

end counter_config;

Die Simulation zeigte die richtige Funktionsweise des Zählers und lieferte folgenden Output für die ersten 1500 ns:

...und für die zweiten 1500 ns:

